

Input the programs as printed without any extra spaces, to check if you have entered the program correctly enter "WORDS" "LIST" "GO", the program should be listed out just like the listing printed here. If you want to shorten Emperor (2) to a fewer number of years change the following line:

Line 34.....IF Y>9GOTO 38 to IF Y>nGOTO 38 n = years - 1

When playing the game enter your name one letter at a time pressing "GO" after entering your name.

When HARVEST ?? BHLS/AC appears, press any key to continue.

KING (1)
by R M Houser (82)

This is a one player economic game using the Keypad for input.

As King you must make decisions on running your kingdom economically for a reign of ten years. You start with 100 people, 1000 acres of land and 3000 bushels of food. Buy and sell land, feed people, and plant crops using food as barter. The object is to complete your reign and garner performance points based on how you govern your kingdom. To get maximum points you need to determine how much you need to feed your people, how many bushels that are required to plant your land, how many acres a person can plant, and a little luck.

>600 pts = Fantastic	>200 pts = Poor
>500 pts = Very Good	>100 pts = Very Poor
>400 pts = Good	<100 pts = Try Again
>300 pts = Fair	

KING (1)
by R M Houser (82)

```

1 CLEAR; PRINT " KING (1)
  BY R M HOUSER";PRINT
2 PRINT " ENTER YOUR FIRS
T NAME";O=0
4 FOR Z=1TO 10;B=KP
6 IF B=13GOTO 10
8 TV=B;O=O+1;E(Z)=B;NEXT
Z
10 CLEAR
20 GOTO 80
30 CY=16;PRINT F," BUSHELS
  OF FOOD";RETURN
40 CY=8;PRINT A," ACRES OF
  LAND";RETURN
50 PRINT " HOW?";RETURN
60 FOR E=1TO 2000;NEXT E;R
  ETURN
70 CY=-24;PRINT "
  ";
  RETURN
80 P=100;F=3000;A=1000;Y=0
  ;L=0;NT=0;W=0;K=0;N=18+RND
  (3)
90 CLEAR
100 PRINT " KING ",;FOR Z=
  1TO 0;TV=@(Z);NEXT Z
110 IF Y>0GOTO 140
120 PRINT " YOU START YOUR
  RULE WITH:
130 GOTO 180
140 IF Y=10GOTO 170
150 PRINT " AFTER ",#1,Y,"
  YEAR(S) OF RULE:
160 GOTO 180
170 PRINT " AT THE END OF
  YOUR REIGN:
180 PRINT P," PEOPLE
190 GOSUB 30;GOSUB 40
200 IF Y#10GOTO 280
210 PRINT K," PEOPLE DIED"
  ;PRINT ;Z=(P-50)+(F÷100)+(
  (A÷100)×20)-((K-10)×2);PRI
  NT Z," PERFORMANCE PTS.
250 IF Z>M M=Z
260 PRINT " TODAY'S HIGH S
  CORE=",#5,M

```

```

270 PRINT " AGAIN?";A=KP;G
  OTO 2
280 V=17+RND (5)
290 PRINT V," BHL/AC LAND
  VALUE
300 PRINT " BUY LAND # OF
  ACRES?
310 INPUT B
320 IF B=0GOTO 410
330 L=B×V
340 IF L<FGOTO 380
350 IF L>FGOSUB 50
360 GOSUB 60;GOSUB 70
370 GOTO 310
380 F=F-L;A=A+B
390 GOSUB 30;GOSUB 40
400 IF B>0GOTO 510
410 CY=-8;PRINT " SELL LAN
  D # OF ACRES?
420 INPUT S
430 IF S=0GOTO 510
440 IF S<0GOTO 490
450 IF S=APRINT " QUITTER"
  ";GOTO 260
460 GOSUB 50
470 GOSUB 60;GOSUB 70
480 GOTO 420
490 A=A-S;F=F+V×S
500 GOSUB 30;GOSUB 40
510 CY=-8;PRINT " GIVE OUT
  FOOD # BUSHELS?
520 INPUT C
530 IF C<FGOTO 560
540 IF C>FGOSUB 50
550 GOSUB 60;GOSUB 70;GOTO
  520
560 F=F-C;GOSUB 30
570 CY=-8;PRINT " PLANT CR
  OPS? # OF ACRES? "
580 GOTO 600
590 GOSUB 60;GOSUB 70
600 INPUT G;X=G÷2
610 IF (G>A)+(X>F)+(G>15×P
  )GOSUB 50;GOTO 590
620 F=F-X;GOSUB 30;GOSUB 6
  0
630 D=P-C÷N
640 IF D>0GOTO 660

```

```

650 IF D<0D=0;U=RND (3)-1
660 H=RND (4)+1;T=H×G
670 R=((F+T)÷100)×RND (7)
680 F=F-R+T;J=RND (9)
690 I=((F÷250)-(D+1))+((A÷
  P)÷10)
700 IF I<=50GOTO 720
710 I=50
720 P=P+I-D+U;Y=Y+1;W=W+1;
  K=K+D;CLEAR
730 PRINT " KING ",;FOR Z=
  1TO 0;TV=@(Z);NEXT Z
740 PRINT " LAST YEAR
750 IF D>P-4IF J#5IF Y>3PR
  INT D," PEOPLE STARVED
  YOU HAVE BEEN THROWN IN PR
  ISON!";GOTO 260
760 IF D>0PRINT D," DEATHS
770 IF U>0PRINT U," BIRTHS
780 PRINT I," PEOPLE IMMI
  GRATED
790 IF W>P÷2+W÷3PRINT " IM
  MIGRANTS EXCEED COUNTRY-ME
  N AND HAVE TAKEN OVER TH
  E COUNTRY";GOTO 260
800 IF J#5GOTO 850
810 PRINT " THE PLAGUE KIL
  LED
820 Q=(P×(10×RND (5)))÷100
830 PRINT Q," PEOPLE
840 P=P-Q
850 PRINT " YOUR PEOPLE HA
  RVESTED
860 PRINT T," BUSHELS AT
870 PRINT H," BUSHELS PER
  ACRE
880 PRINT " RATS ATE ",#3,
  R," BUSHELS
890 GOSUB 60;GOTO 90

```

EMPEROR (2)
by R M Houser (82)

This is a two player economic game using the keypad for input.

The present emperor of the Kingdom of Astro is getting along in years and has decided to have a contest between two of his Lords to determine who will be selected the next emperor. Each lord will start with 100 people, 3000 bushels of food, and 1000 acres of land. During a period of 10 years each lord must govern his allotment and at the end of ten years will be awarded performance points based on how many people, how many bushels of food, how many acres of land. The Lord with the most points will be selected the next emperor.

EMPEROR (2) by R M HOUSER

```

1 K=1;N=0;O=0
2 CLEAR;GOTO 18
3 PRINT " LORD ",;RETURN
4 CX=-71;RETURN
5 CY=-16;RETURN
6 CY=-32;RETURN
7 CX=-71;CY=-24;RETURN
8 CX=-77;CY=16;PRINT #5,P
," PEOPLE,NO. ",0;PRINT #
5,F," FOOD,BHLS. ",G;PRIN
T #5,L," LAND,ACRES ",M;R
ETURN
9 GOSUB 6;PRINT " HOW?";F
OR A=1TO 2000;NEXT A;GOSUB
6;PRINT " ";GOSUB
6;RETURN
10 CLEAR ;PRINT " LORD
LORD";GOSUB 4;GOSUB
14;CX=12;GOSUB 16;RETURN
12 PRINT #5,U," BHL/AC LAN
D VALUE";RETURN
14 FOR A=10TO 9+N;TV=@(A);
NEXT A;RETURN
16 FOR A=20TO 19+0;TV=@(A)
;NEXT A;RETURN
18 PRINT " PLAYER ",K;PRIN
T " ENTER FIRST NAME
20 FOR A=(K*10)TO (K*10)+1
0;W=KP
22 IF W=13K=K+1;IF K=2GOTO
2
24 IF K>2GOTO 30
26 TV=W;IF K=1N=N+1;@(A)=W
;NEXT A
28 IF K=2O=O+1;@(A)=W;NEXT
A
30 CLEAR ;P=100;Q=P;F=3000
;G=F;L=1000;M=L;Y=0;NT=0;Z
=15+RND (9);D=0;E=0;GOSUB
10
32 CY=24;GOSUB 4;PRINT " S
TART TRIAL PERIOD";GOTO 40
34 IF Y>9GOTO 38
36 CY=24;GOSUB 4;PRINT "AF
TER ",#1,Y," YEAR(S)";GOTO
40
38 GOSUB 10;CY=24;GOSUB 4;
PRINT "END TRIAL PERIOD
40 GOSUB 8
42 IF Y#10GOTO 54
44 PRINT #5,D," DEATHS
",E;B=P-50+(F/99)+L/5-((
D-9)*2)
46 C=Q-50+(G/99)+M/5-((E-9
)*2);PRINT #5,B," PERF. PT
S. ",C
48 GOSUB 7;GOSUB 3;IF B>CG
OSUB 14;GOTO 52
50 GOSUB 16
52 PRINT " SELECTED";GOSUB
6;CX=-77;PRINT " EMPEROR"
;STOP
54 U=15+RND (9);GOSUB 12;K
=0;GOSUB 3;GOSUB 14
56 GOSUB 7;PRINT "BUY LAND
?"
58 INPUT A;X=A*U;IF K<GOTO
64
60 IF X>FGOSUB 9;GOTO 58
62 IF X<F F=F-X;L=L+A;GOTO
68
64 IF X>GGOSUB 9;GOTO 58
66 IF X<G G=G-X;M=M+A
68 GOSUB 10;GOSUB 8;GOSUE
12;K=K+1;IF K=1GOSUB 5;GOS
UB 3;GOSUB 16;GOTO 56
70 K=0;GOSUB 3;GOSUB 14
72 GOSUB 7;PRINT "SELL LAN
D?"
74 INPUT A;X=A*U;IF K<GOTO
80
76 IF A>L B=0;C=1;CLEAR ;
GOTO 48
78 F=F+X;L=L-A;GOTO 84
80 IF A>M C=0;B=1;CLEAR ;
GOTO 48
82 G=G+X;M=M-A
84 GOSUB 10;GOSUB 8;GOSUB
12;K=K+1;IF K=1GOSUB 5;GOS
UB 3;GOSUB 16;GOTO 72
86 GOSUB 3;GOSUB 14;GOSUB
7;PRINT "GIVE OUT FOOD?"
88 INPUT U;IF U>F GOSUB 9;
GOTO 88
90 IF U<F F=F-U;GOSUB 8
92 GOSUB 5;GOSUB 3;GOSUB 1
6
94 GOSUB 6;CX=-77;INPUT W
96 IF W>G GOSUB 9;GOTO 94
98 IF W<G G=G-W;GOSUB 8
100 K=1;GOSUB 5;GOSUB 3;GO
SUB 14
102 GOSUB 7;PRINT "PLANT C
ROPS?"
104 IF K=1;GOSUB 6;INPUT T
;X=T/2;IF (T>L)+(X>F)+(T>1
5*X)GOSUB 9;GOTO 104
106 IF K=2;GOSUB 6;INPUT U
;X=U/2;IF (U>M)+(X>G)+(U>1
5*X)GOSUB 9;GOTO 106
108 IF K=1F=F-X;GOSUB 8
110 IF K=2G=G-X;GOSUB 8
112 K=K+1;IF K=2GOSUB 5;GO
SUB 3;GOSUB 16;GOTO 102
114 I=P-U/2;B=RND ((P+20)/
20)
116 J=Q-W/2;C=RND ((Q+20)/
20)
118 K=RND (4)+1;R=K*T;S=K*
U;P=P-I+B;Q=Q-J+C;X=R/99*R
ND (3);H=S/99*RND (3);F=F+
R-X;G=G+S-H;CLEAR ;GOSUB 1
0;GOSUB 8
120 PRINT " HARVEST",K," B
HL/AC";IF (P<10)+(Q<10)CLE
AR ;GOTO 44
122 Y=Y+1;D=D+I;E=E+J;A=KP
;GOTO 34

```

Hints for playing KING (1) and EMPEROR (2).

1. For the whole period of ten years the amount of food required by each person remains the same. It is somewhere above 10 bushels per person.
2. A person can only plant so many acres. It is less than 20 acres per person. Plant as many acres as you can.
3. The value of land in bushels per acre changes each year.
4. Sell land when land value is high and buy when land value is low.
5. Rats eat some of the food each year.
6. There are a random number of births each year based on the total population during the year.
7. If either population becomes lower than 10, the other Lord wins.
8. If you sell all of your land, you lose to the other Lord.
9. In case of ties, the 2nd Lord wins.
10. The highest score that I have made in Emperor is 569 points.